
Dit al enige tijd geleden ingezonden programma zet tekstbestanden, gemaakt met Home 
Office 2, om in gewone ASCII-files, zodat ze voor verscheidene andere toepassingen 
gebruikt kunnen worden.  
De eenvoudigste manier om teksten op diskette op te slaan, is als zogenaamd ASCII-
bestand. De enige informatie die hierin wordt opgeslagen, is de werkelijke tekst. Omdat 
bijna alle programma's met dergelijke bestanden overweg kunnen/ zijn ASCII-teksten 
voor allerlei toepassingen bruikbaar. De eenvoud van deze files heeft echter ook een 
nadeel.  
Voor tekstverwerkers is er geen mogelijkheid andere gegevens over de tekst op te slaan/ 
zoals welke kantlijnen worden gebruikt. Daarom maken programma's meestal een eigen 
type file/ waarin voor dit alles wel een plaats is. Meestal kan er echter/ als de gebruiker 
dat wil/ toch een simpel ASCII-bestand met alleen de tekst bewaard worden, 
Ook de tekstverwerker in het Philips-programma Home Office 2 gebruikt een eigen soort 
bestand. Het is herkenbaar aan de extensie .CWP. Vanuit dit pakket kan een tekst helaas 
echter niet zonder allerlei toevoegingen worden bewaard, zodat teksten bijvoorbeeld niet 
gebruikt kunnen worden als programmalisting of in een andere tekstverwerker. Alleen het 
eigen bestandsformaat kan gebruikt worden. Een actieve Veghelse lezer van MSX Club 
Magazine, zond de redactie, alweer enige tijd geleden  HOME2ASC, een programma, 
waarmee tekstfiles van Home Office 2 vertaald kunnen worden in pure ASCII-bestanden. 
 
Listing 
 
Het korte programma is geschreven in Turbo Pascal. Om de listing, die in het kader naast 
dit artikel te vinden is, zelf in te kunnen voeren/ dient u dan ook in het bezit te zijn van 
een compiler voor deze taal. Nadat de listing in de editor is ingevoerd, maakt de compiler 
de file HOME2ASC.COM aan. Voor degenen die Turbo Pascal niet bezitten of er geen 
zin in hebben de listing in te moeten typen, is het kant en klare programma ook op de 
diskette bij dit blad te vinden. 
Gebruik van het programma 
Nadat het programma gecompileerd is —of van de disk is gehaald—kan het vanuit MSX-
DOS gestart worden door de naam in te voeren: 
 
HOME2ASC 
 
Het programma vraagt allereerst om de naam van het Home Office-bestand dat omgezet 



moet worden. Voer hierbij de volledige naam in, inclusief extensie en met een eventuele 
drivenaam ervoor, zoals: 
 
B:TEST.CWP 
 
Vervolgens vraagt HOME2ASC naar de naam van de te produceren ASCII-file. Dit moet 
een nog niet voorkomende file zijn, bijvoorbeeld: 
 
C:TEST.ASC 
 
Wanneer er bij één van beide bestandsnamen geen drivenaam wordt vermeld, wordt 
uiteraard de standaard-drive gebruikt. 
 
Tenslotte beschikt dit programma over de mogelijkheid om indien er meerdere spaties 
achter elkaar staan deze op één na allemaal weg te halen. Dit komt voor wanneer de 
oorspronkelijke tekst is uit-gevuld. Deze optie kan ingeschakeld worden door achter de 
naam van het programma een grote of een kleine 'U' op te nemen: 
 
HOME2ASC U 
 
Snelheid en werking 
 
Home Office to ASCII is in Turbo Pascal geschreven omdat deze gecompileerde 
programma's met hoge snelheid uitgevoerd kunnen worden. Een programma met dezelfde 
werking zou in BASIC veel meer tijd nodig hebben. Een document van bijvoorbeeld 115 
kB vraagt nu rond de tien minuten om omgezet te worden; onder BASIC zou daarvoor 
een half uur verloren gaan. Wanneer tijdens het vertalen een ramdisk voor de files 
gebruikt wordt  loopt de genoemde tijd zelfs tot slechts drie minuten terug. 
De codes en toevoegingen die door het programma uit de Home Office-teksten worden 
verwijderd  zijn de volgende: 
• de eerste karakters van het bestand/ afgesloten door een byte FFh  omdat deze alleen 

voor Home Office 2 zelf van belang zijn; 
• de tabs  omdat deze alleen voor de inspring-functie gebruikt worden; 
• e printercodes; 
• de kopregels; 
• eventueel meerdere spaties achter elkaar omdat deze zoals hiervoor besproken is  

gebruikt worden voor het uitvullen van regels. 
 
Daarnaast wordt de door Home Office gebruikte code voor het einde van een alinea een 
byte met de waarde 141  vervangen door de standaard ASCII-codes carriage return en 
linefeed (cr lf/13 10). 
 



{$V-,C-} program home2asc;    {version 2.0} 
 
{Dit programma zet een Home Office 2 tekstfile 
 om in een ASCII-file} 
 
type 
  string14=string[14]; 
  string20=string[20]; 
  string100=string[100]; 
 
var 
  oud,nieuw:text; 
  kar1,kar:char;  
  oudenaam,nieuwenaam:string14; 
  bkopregel,ekopregel:string20; 
  uitvullen,foutmelding1,foutmelding2:string100; 
  spaties,uitv,kode,io:integer; 
 
procedure kop; 
  begin 
    writeln; 
    writeln('V-Soft    Home Office 2 -> ASCII v.2.0'); 
    writeln('--------------------------------------'); 
end; {kop} 
 
{Nu worden twee files geopend. Er wordt gekeken of 
 de file die moet worden omgezet al op schijf staat. 
 Zo niet, stopt het programma, net als wanneer de 
 aan te maken file juist wel bestaat.} 
 
procedure openfiles; 
 
begin 
  foutmelding1:=' '; 
  foutmelding2:=' '; 
  io:=0; 
  write('Geef naam van de Home Office file:  '); 
  readln(oudenaam); 
  write('Geef naam van de nieuwe ASCII-file: '); 
  readln(nieuwenaam); 
  assign(oud,oudenaam); 
  {$I-} reset(oud); {$I+}      {open file voor lezen} 
  if ioresult=1 then           {oude file niet op disk} 
  begin 
    io:=1; 
    foutmelding1:='Home Office file niet gevonden.'; 
  end; 
  assign(nieuw,nieuwenaam); 
  {$I-} reset(nieuw); {$I+} 
  if ioresult<>1 then          {nieuwe file bestaat al} 
  begin 
    io:=2; 
    foutmelding2:='ASCII file bestaat al.'; 
  end; 
  if io=0 then 
  rewrite(nieuw);              {open file voor schrijven} 
end; {openfiles} 



 
{Onderstaande procedure verwijdert de eerste karakters 
 van een Home Office file met bepaalde onnodige codes, 
 tot een karakter $FF. 
} 
 
procedure weghalen; 
var dummy:integer; 
begin 
  while kar<>chr($FF) do 
  begin 
    read(oud,kar); 
  end; {while} 
  read(oud,kar); 
end; {weghalen} 
 
{Procedure 'tabs' haalt de tabcodes weg. Dit zijn geen 
 gewone tabs, maar worden in Home Office alleen gebruikt 
 aan de linker kantlijn. 
} 
 
procedure tabs; 
var dummy:integer; 
begin 
  kode:=1; 
  read(oud,kar); 
end; {tabs} 
 
{De procedure 'kop-printercode' verwijdert kopregels en 
 printercodes uit de tekst. 
} 
 
procedure kop_printercode; 
begin 
  kode:=1; 
  read(oud,kar); 
  if kar=chr($54) then         {begin kopregel} 
  begin 
    read(oud,kar); 
    while kar<>chr($54) do 
    begin 
      read(oud,kar); 
    end; {while}               {einde kopregel} 
    read(oud,kar); 
  end; {if} 
end; {kopregel} 
 
{Onderstaande procedure kopieert tekens van 32 tot 127 
 naar de ASCII-file. 
} 
 
procedure schrijf; 
begin 
  if uitv=1 then 
  begin                        {voor de U-optie} 
    if (spaties=0) or ((spaties=1) and (kar<>chr(32))) then 
    begin 



      if (kar>=chr(32)) and (kar<chr(127)) then 
      begin 
        write(nieuw,kar); 
        kode:=0; 
        if kar=chr(32) then spaties:=1 else spaties:=0; 
      end; {if} 
    end; {if spaties} 
  end {if uitv=1} 
  else 
  begin                        {niet voor de U-optie} 
    if (kar>=chr(32)) and (kar<chr(127)) then 
    begin 
      write(nieuw,kar); 
      kode:=0; 
    end; {if kar} 
  end; {if uitv=0} 
end; {schrijf} 
 
{Procedure 'crlf' verandert het karakter 141 in een cr-lf. 
} 
 
procedure crlf; 
begin 
  kode:=1; 
  write(nieuw,chr(13)+chr(10)); 
end; {crlf} 
 
{Onderstaande procedure controleert de tekst op de diverse 
 codes en coordineert de bijbehorende procedures. 
} 
 
procedure controle; 
begin 
  spaties:=0; 
  while not eof(oud) do 
  begin 
    kode:=0; 
    read(oud,kar); 
    if kar=chr(1) then kop_printercode; 
    if kar=chr(9) then tabs; 
    if kar=chr(141) then crlf; 
    if kar>chr(127) then kar:=chr(ord(kar)-128); 
    if kode=0 then schrijf; 
  end; {while not eof} 
end; {controle} 
 
{Hoofdprogramma} 
 
begin 
  uitv:=0; 
  if paramcount<>0 then 
  begin 
    uitvullen:=paramstr(1); 
    kar1:=copy(uitvullen,1,1); 
    if (kar1='U') or (kar1='u') then uitv:=1; 
  end; {if paramcount} 
  kop; 



  openfiles; 
  if io=0 then 
  begin 
    writeln; 
    writeln('Even geduld aub...'); 
    weghalen; 
    controle; 
  end; {if io} 
  if io=1 then writeln(foutmelding1); 
  if io=2 then writeln(foutmelding2); 
  close(oud); 
  close(nieuw); 
end. 

